SCENES/scripts/SCENE_QUERY.BAT

@ECHO OFF
CHCP 65001
COLOR 0A
TITLE SCENES EXTRACTOR
::
SET USELOCALPATH=0
SET USENEWFILES=0
::
IF %USELOCALPATH% EQU 0 (
	SET "TARGET_PATH=Q:\PSHOME-INT\PSHomeCacheDepot\SCENES"
) ELSE (
	SET "TARGET_PATH=%SRC1%SCENES"
)
SETLOCAL ENABLEEXTENSIONS DISABLEDELAYEDEXPANSION
SET "SRC1=%~DP0"
SET BIN=%~DP0BIN\
SET "DUPES=log_DUPE_INF_PATH.log"
SET "NEWFILES=NewFiles.txt"
SET "SCENEQUERY=SCENE_QUERY.TXT"
IF %USENEWFILES% EQU 0 (
	SET "QUERYLIST=%SRC1%%SCENEQUERY%"
	SET "FILENAME=%SCENEQUERY%"
) ELSE (
	SET "QUERYLIST=%SRC1%%NEWFILES%"
	SET "FILENAME=%NEWFILES%"
)
SET "EXT[0]=sdat"
SET "EXT[1]=sdc"
SET "EXT[2]=png"
SET "EXT[3]=bar"

SET "LOCALPATH=R:\PSHOME\CACHE\BIN"

SET "EXE=%LOCALPATH%\PSHomeCacheTools\file_analysis.exe"
SET "DLL1=%LOCALPATH%\PSHomeCacheTools\python311.dll"
SET "LIB=%LOCALPATH%\PSHomeCacheTools\lib"
IF NOT EXIST "%LOCALPATH%" ECHO BIN FOLDER NOT FOUND @ %LOCALPATH% & GOTO END
SETLOCAL ENABLEDELAYEDEXPANSION
SET "3RDPARTYFILES="
IF NOT EXIST "%EXE%" SET "3RDPARTYFILES=!3RDPARTYFILES! file_analysis.exe"
IF NOT EXIST "%DLL1%" SET "3RDPARTYFILES=!3RDPARTYFILES! python311.dll"
IF NOT EXIST "%LIB%" SET "3RDPARTYFILES=!3RDPARTYFILES! lib"
IF NOT "!3RDPARTYFILES!" == "" (
    ECHO DEPENDENCIES NOT FOUND AT %LOCALPATH%
    FOR %%F IN (!3RDPARTYFILES!) DO (
        ECHO    %%F
    )
    GOTO END
)
ENDLOCAL


SET INDEX=0

:INDEX_LOOP

IF DEFINED EXT[%INDEX%] (
	SET /A INDEX+=1
	GOTO :INDEX_LOOP
)

:MENU_MAIN
ECHO.
ECHO.^|--------------------^|
ECHO.^|   SCENES OPTIONS   ^|
ECHO.^|--------------------^|
ECHO.
ECHO 1. QUERY SCENES
ECHO 2. COPY SCENES
ECHO 3. SUSPEND CMD
ECHO.
SET /P CHOICE=ENTER A NUMBER TO PERFORM AN ACTION: 
ECHO.
IF NOT '%CHOICE%'=='' SET CHOICE=%CHOICE:~0,1%
IF '%CHOICE%'=='1' GOTO MAIN_CHOICE_ONE
IF '%CHOICE%'=='2' GOTO MAIN_CHOICE_TWO
IF '%CHOICE%'=='3' GOTO MAIN_CHOICE_THREE

ECHO "%CHOICE%" IS NOT VALID, TRY AGAIN
ECHO.
GOTO MENU_MAIN

:MAIN_CHOICE_ONE
CALL :SCENES_QUERY
GOTO MAIN_OPTION_END

:MAIN_CHOICE_TWO
CALL :SCENES_COPY
GOTO MAIN_OPTION_END

:MAIN_CHOICE_THREE
CMD /K
GOTO MAIN_OPTION_END

:MAIN_OPTION_END
CMD /K
GOTO MENU_MAIN


:SCENES_QUERY

IF %USENEWFILES% EQU 1 ECHO USENEWFILES NOT SET TO 0 & GOTO MENU_MAIN

IF EXIST "%SRC1%scee" REN "%SRC1%scee" "scee-home.playstation.net"

IF EXIST "%QUERYLIST%" BREAK > "%QUERYLIST%"

CALL :STARTTIMER

FOR /L %%N IN (0, 1, %INDEX%) DO (
	@ECHO OFF
	CALL SET Z1=%%EXT[%%N]%%
	SET COUNTER=0
	FOR /F "TOKENS=*" %%A IN ('DIR /B /A:D "%SRC1%\*" ^| FINDSTR /V /RC:"\<BIN\>"') DO FOR /F "TOKENS=*" %%B IN ('DIR /B /A:-D /S "%SRC1%%%A\*.%%Z1%%" 2^>NUL ^| FINDSTR /I \\Scenes\\') DO (
		SET "A=%%A"
		SET "B=%%B"
		SET "C=%%~NXB"
		SET "D=%%~DPB"
		SET "E=%%~NB"
		SET "EXT=%%~XB"
		SET /A COUNTER+=1
		CALL :CONT_QUERY_SCENES
	)
)

CALL :STOPTIMER
CALL :DISPLAYTIMERRESULT
GOTO MENU_MAIN
CMD /K

:CONT_QUERY_SCENES

CALL SET "A=%%A%%"
CALL SET "B=%%B%%"
CALL SET "C=%%C%%"
CALL SET "D=%%D%%"
CALL SET "E=%%E%%"
CALL SET "F=%%D:~0,-1%%"
CALL SET "H=%%F:%SRC1%=%%"
CALL SET "EXT=%%EXT%%"

SET "SDATNAME=%H:\=" & REM."%"
SET "MAINFOLDER=%H:\=" & SET "MAINFOLDER=%"


SETLOCAL ENABLEDELAYEDEXPANSION

SET "H=!H:corrupted\=!"

CALL :GET_FILE_INFO "!B!"

SET "FILEEXE=!FILEEXT!"
SET "FILESIZE=!FILESIZE!"
SET "SHA1=!SHA1HASH!"
SET "SDATVER=!SDATVER!"
SET "MISSINGBYTES=!MISSINGBYTES!"

>>"%QUERYLIST%" (
	(
	ECHO !FILESIZE!	!SDATVER!	!H!\!C!
	)
)

IF !SDATVER! EQU 0 (
	IF EXIST "%SRC1%corrupted\!H!\!C!" (
		BREAK
	) ELSE (
		ECHO F|XCOPY "!B!" "%SRC1%corrupted\!H!\!C!" /Y >NUL
		ECHO !H!\!C! - CORRUPT
		IF EXIST "%SRC1%!H!\!C!" DEL "%SRC1%!H!\!C!"
	)
)

ECHO !COUNTER! LOGGED !H!\!C!

ENDLOCAL
EXIT /B

:SCENES_COPY
SETLOCAL ENABLEDELAYEDEXPANSION

IF NOT EXIST "%QUERYLIST%" ECHO %FILENAME% NOT FOUND & GOTO MENU_MAIN

IF NOT EXIST "%TARGET_PATH%" (
	MKDIR "%TARGET_PATH%" || (
		ECHO %TARGET_PATH% NOT FOUND
		GOTO MENU_MAIN
	)
	ECHO CREATED %TARGET_PATH%
)

CALL :STARTTIMER

IF EXIST "%SRC1%%DUPES%" BREAK > "%SRC1%%DUPES%"

IF %USENEWFILES% EQU 0 (
	IF EXIST "%SRC1%%NEWFILES%" BREAK > "%SRC1%%NEWFILES%"
)

FOR /F "TOKENS=1,2,3 USEBACKQ DELIMS=	" %%A IN ("%QUERYLIST%") DO (
	SET "A=%%A"
	SET "B=%%B"
	SET "C=%%C"
	
	IF !B! EQU 0 (
	
		CALL :GET_FILE_INFO "%SRC1%corrupted\!C!"
		
		SET "NEW_FILEEXT=!FILEEXT!"
		SET "NEW_FILESIZE=!FILESIZE!"
		SET "NEW_SHA1=!SHA1HASH!"
		SET "NEW_SDATVER=!SDATVER!"
		SET "NEW_MISSINGBYTES=!MISSINGBYTES!"
		
	) ELSE IF !B! EQU 3.3 (
	
		CALL :GET_FILE_INFO "%SRC1%modified\!C!"
		
		SET "NEW_FILEEXT=!FILEEXT!"
		SET "NEW_FILESIZE=!FILESIZE!"
		SET "NEW_SHA1=!SHA1HASH!"
		SET "NEW_SDATVER=!SDATVER!"
		SET "NEW_MISSINGBYTES=!MISSINGBYTES!"
	
	) ELSE (
	
		CALL :GET_FILE_INFO "%SRC1%!C!"
		
		SET "NEW_FILEEXT=!FILEEXT!"
		SET "NEW_FILESIZE=!FILESIZE!"
		SET "NEW_SHA1=!SHA1HASH!"
		SET "NEW_SDATVER=!SDATVER!"
		SET "NEW_MISSINGBYTES=!MISSINGBYTES!"

	)
	
	IF !B! EQU 0 (

		IF EXIST "%TARGET_PATH%\corrupted\!C!" (
			
		  CALL :GET_FILE_INFO "%TARGET_PATH%\corrupted\!C!"
		  
		  SET "OLD_FILEEXT=!FILEEXT!"
		  SET "OLD_FILESIZE=!FILESIZE!"
		  SET "OLD_SHA1=!SHA1HASH!"
		  SET "OLD_SDATVER=!SDATVER!"
		  SET "OLD_MISSINGBYTES=!MISSINGBYTES!"
			
			IF !NEW_FILESIZE! GTR !OLD_FILESIZE! (
				ECHO F|XCOPY "%SRC1%corrupted\!C!" "%TARGET_PATH%\corrupted\!C!" /Y >NUL
				ECHO COPIED !C! - NEWER CORRUPT SDAT ^( !NEW_FILESIZE! ^)
			) ELSE (
				(ECHO corrupted\!C!)>>"%SRC1%%DUPES%"
			)

		) ELSE (
			ECHO F|XCOPY "%SRC1%corrupted\!C!" "%TARGET_PATH%\corrupted\!C!" /Y >NUL
			ECHO COPIED !C! - CORRUPT SDAT
		) 

	) ELSE IF !B! EQU 3.3 (
	
		IF EXIST "%TARGET_PATH%\modified\!C!" (
			
		  CALL :GET_FILE_INFO "%TARGET_PATH%\modified\!C!"
		  
		  SET "OLD_FILEEXT=!FILEEXT!"
		  SET "OLD_FILESIZE=!FILESIZE!"
		  SET "OLD_SHA1=!SHA1HASH!"
		  SET "OLD_SDATVER=!SDATVER!"
		  SET "OLD_MISSINGBYTES=!MISSINGBYTES!"
			
			IF !NEW_FILESIZE! GTR !OLD_FILESIZE! (
				ECHO F|XCOPY "%SRC1%modified\!C!" "%TARGET_PATH%\modified\!C!" /Y >NUL
				ECHO COPIED !C! - NEWER MODIFIED SDAT ^( !NEW_FILESIZE! ^)
			) ELSE (
				(ECHO modified\!C!)>>"%SRC1%%DUPES%"
			)

		) ELSE (
			ECHO F|XCOPY "%SRC1%modified\!C!" "%TARGET_PATH%\modified\!C!" /Y >NUL
			ECHO COPIED !C! - MODIFIED SDAT
		) 
	
	) ELSE (
	
		IF !NEW_FILESIZE! EQU 0 (

			IF EXIST "%TARGET_PATH%\corrupted\!C!" (
				(ECHO corrupted\!C!)>>"%SRC1%%DUPES%"
			) ELSE (
				ECHO F|XCOPY "%SRC1%!C!" "%TARGET_PATH%\corrupted\!C!" /Y >NUL
				ECHO COPIED !C! - CORRUPT
			)

		) ELSE (
	
			IF EXIST "%TARGET_PATH%\!C!" (
			
				IF /I !NEW_FILEEXT! == .sdat (
				
			  	CALL :GET_FILE_INFO "%TARGET_PATH%\!C!"
			  
			  	SET "OLD_FILEEXT=!FILEEXT!"
			  	SET "OLD_FILESIZE=!FILESIZE!"
			  	SET "OLD_SHA1=!SHA1HASH!"
			  	SET "OLD_SDATVER=!SDATVER!"
			  	SET "OLD_MISSINGBYTES=!MISSINGBYTES!"
    	
					IF !NEW_FILESIZE! GTR !OLD_FILESIZE! (
						ECHO F|XCOPY "%SRC1%!C!" "%TARGET_PATH%\!C!" /Y >NUL
						ECHO COPIED !C! - NEWER SDAT ^( !NEW_FILESIZE! ^)
            IF %USENEWFILES% EQU 0 (
            	(ECHO !NEW_FILESIZE!	!NEW_SDATVER!	!C!)>>"%SRC1%%NEWFILES%"
            )
					) ELSE (
						(ECHO !C!)>>"%SRC1%%DUPES%"
					)
				) ELSE (
					(ECHO !C!)>>"%SRC1%%DUPES%"
				)
			) ELSE (
				ECHO F|XCOPY "%SRC1%!C!" "%TARGET_PATH%\!C!" /Y >NUL
        IF %USENEWFILES% EQU 0 (
        		(ECHO !NEW_FILESIZE!	!NEW_SDATVER!	!C!)>>"%SRC1%%NEWFILES%"
        )
				ECHO COPIED !C!
			)
		)
	)
)

IF EXIST "%SRC1%%DUPES%" SORT /UNIQUE "%SRC1%%DUPES%" /O "%SRC1%%DUPES%"

CALL :STOPTIMER
CALL :DISPLAYTIMERRESULT

ENDLOCAL

GOTO MENU_MAIN

:STARTTIMER
:: STORE START TIME
SET STARTTIME=%TIME%
FOR /F "USEBACKQ TOKENS=1-4 DELIMS=:., " %%F IN (`ECHO %STARTTIME: =0%`) DO SET /A START100S=1%%F*360000+1%%G*6000+1%%H*100+1%%I-36610100
GOTO :EOF

:STOPTIMER
:: GET THE END TIME
SET STOPTIME=%TIME%
FOR /F "USEBACKQ TOKENS=1-4 DELIMS=:., " %%F IN (`ECHO %STOPTIME: =0%`) DO SET /A STOP100S=1%%F*360000+1%%G*6000+1%%H*100+1%%I-36610100
IF %STOP100S% LSS %START100S% SET /A STOP100S+=8640000
SET /A TOOKTIME=%STOP100S% - %START100S%
SET TOOKTIMEPADDED=0%TOOKTIME%
SET /A TOOKSECS=%TOOKTIME% / 100
SET /A SECS=%TOOKSECS% %% 60
SET /A MINS=(%TOOKSECS% / 60) %% 60
SET /A HOURS=%TOOKSECS% / 3600
IF %SECS% LSS 10 SET SECS=0%SECS%
IF %MINS% LSS 10 SET MINS=0%MINS%
IF %HOURS% LSS 10 SET HOURS=0%HOURS%
GOTO :EOF

:DISPLAYTIMERRESULT
:: SHOW TIMER START/STOP/DELTA
ECHO STARTED: %STARTTIME%
ECHO STOPPED: %STOPTIME%
ECHO ELAPSED : %HOURS%h:%MINS%m:%SECS%s
GOTO :EOF

:LOWERCASE <RETURN_VAR> <STRING>
SETLOCAL DISABLEDELAYEDEXPANSION
IF "%~2"=="" (SET RET=0) ELSE SET RET=1
SET "TMPSTR=%~2"
FOR %%A IN ("A=a" "B=b" "C=c" "D=d" "E=e" "F=f" "G=g" "H=h" "I=i" "J=j" "K=k" "L=l" "M=m" "N=n" "O=o" "P=p" "Q=q" "R=r" "S=s" "T=t" "U=u" "V=v" "W=w" "X=x" "Y=y" "Z=z" "�=�" "�=�" "�=�") DO (
CALL SET "TMPSTR=%%TMPSTR:%%~A%%
)
ENDLOCAL & SET "%~1=%TMPSTR%"
GOTO :EOF

:GET_FILE_INFO
FOR /F "TOKENS=1,2,3,4,5 DELIMS=	" %%K IN ('CALL "%EXE%" "%~1"') DO (
    SET "FILEEXT=%%K"
    SET "FILESIZE=%%L"
    SET "SDATVER=%%M"
    SET "SHA1HASH=%%N"
    SET "MISSINGBYTES=%%O"
)
GOTO :EOF